3 Datenstrukturen
Neben den unstrukturierten Datentypen (Standard-Typen und Aufzählungstypen) gibt es die strukturierten (konstruierten) Typen wie Array (auch string), Record, Set und File. Aus der Sicht der Anwendungsprogrammierung geht es dabei darum, wie Daten im Hauptspeicher angeordnet werden können und wie sie auf Festplatte etc. abgespeichert und wiederge​wonnen werden können (eigentliche „Datenverarbeitung“).

3.1 Arrays (Felder, Matrizen, Tabellen)

Beispiel einer Problembeschreibung: Für eine statistische Berechnung ist es erforderlich, dass viele (z. B. 500) Zahlenwerte eingegeben werden und dass diese Werte für eine nach der Eingabe erforderlichen Berechnung alle im Hauptspeicher des Computers verfügbar sein müssen. Es ist einsichtig, dass eine Deklaration eines jeden einzelne Wertes sehr umständlich wäre, z. B.:

Var

    Z1 : real;

    Z2 : real;

    Z3 : real;

    Z4 : real;

    Z5 : real;

    Z6 : real;

    u.s.w

     Z500 : real;
Für Problemstellungen dieser Art existiert eine wesentlich komfortablere Art der Daten​definition:

Var Z : Array [1..500] of real;

Gleiche Datentypen werden so zu sogenannten Datenfeldern (Arrays, Tabellen) zusammen​gefasst. Die einzelnen Elemente eines Arrays haben alle den gleichen Namen (hier: Z); unterscheiden lassen sie sich durch einen Index, der hinter dem Datennamen in Klammern steht. Das besondere an dem Index ist, dass es sich hier auch um eine Variable handeln kann. Arrayverarbeitungen bzw. Zugriffe auf Arrayfelder können mit Hilfe von Schleifen so sehr geschickt formuliert werden. Bei der Deklaration wird die Anzahl der Datenelemente in der Klammer angegeben (hier: Untergrenze: 1, Obergrenze: 500). 
Die folgende Abbildung veranschaulicht die Speicherung eines Arrays im Hauptspeicher:

Ein kleines Beispiel, hier aber mit String-Variablen als Basis-Datentyp, dient der Einführung in die Array-Thematik:

program Wochentag;

var w_name: Array[1..7] of String[11];

    i : Integer;

Begin

    w_name[1] := 'Montag';

    w_name[2] := 'Dienstag';

    w_name[3] := 'Mittwoch';

    w_name[4] := 'Donnerstag';

    w_name[5] := 'Freitag';

    w_name[6] := 'Samstag';

    w_name[7] := 'Sonntag';

    write('Eingabe Wochentag 1-7: ');

    readln (i);

    writeln('Tag: ',  w_name[i]);

    readln;

End.

Aufgabe 61:
Testen Sie das Programm und erklären Sie bitte seine Funktionsweise! Was passiert, wenn als Wochentag eine Zahl größer 7 oder kleiner 1 eingegeben wird? Korrigieren Sie das Programm mit Hilfe einer geeigneten Abfrage und der Ausgabe einer Fehlermeldung bei falscher Eingabe!

Zugriff auf Arrayfelder in einer Schleife:

program Wochentag;

var w_name: Array[1..7] of String[11];

    i : Integer;

Begin

    w_name[1] := 'Montag';

    w_name[2] := 'Dienstag';

    w_name[3] := 'Mittwoch';

    w_name[4] := 'Donnerstag';

    w_name[5] := 'Freitag';

    w_name[6] := 'Samstag';

    w_name[7] := 'Sonntag';

    for i:= 1 to 7 do

      writeln(i:2,'.Tag: ',  w_name[i]);

    readln;

End.

Aufgabe 62:
Testen Sie das Programm und erstellen Sie bitte das entsprechende Struktogramm! Erklären Sie an diesem Beispiel den oben stehenden Satz: „Arrayverarbeitungen bzw. Zugriffe auf Arrayfelder können mit Hilfe von Schleifen (...) sehr geschickt formuliert werden.“

Problemstellung: Bei einer Qualitätsmessung (12 Stichproben) soll die Abweichung vom Mittelwert errechnet und ausgegeben werden.

Lösung:

program Abweichung_vom_Mittelwert;

var wert: Array[1..12] of real;

    i : Integer;

    g : real;

    m : real;

Begin

    g := 0;

    for i:= 1 to 12 do

    begin

         write(i:2,'.Messung: ');

         readln(wert[i]);

         g := g + wert[i];

    end;

    m := g / 12;

    writeln;

    for i:= 1 to 12 do

    begin

        write('Abweichung vom Mittelwert, ',i:2);

        writeln('.Messung: ', abs(wert[i]-m):8:2);

    end;

    readln;

End.
Aufgabe 63:
Testen Sie das Programm und erstellen Sie bitte das entsprechende Struktogramm! Warum ist dieses Beispiel besonders gut geeignet, den Sinn und den Vorteil von Arrays zu demonstrieren?
Aufgabe 64:
Erweitern Sie das Programm Abweichung_vom_Mittelwert so, dass der Mittelwert angezeigt wird und die Ausgabe  auf dem Bildschirm in Tabellenform erfolgt (die Eingabe bleibt wie oben).

Mittelwert: XXXX.XX

Messung, Nr
Messung, Wert
Abweichung vom Mittelwert

--------------------------------------------------------

 1



XXX.XX



XXX.XX

 2



XXX.XX



XXX.XX

 3 



XXX.XX



XXX.XX

usw.

Aufgabe 65 (nach HERTD):
Eine Fußballmannschaft berechnet jährlich seine Tore mit einem Programm mit folgendem Bildschirmaufbau:

----- Gesamttore der letzten Saison -----

Spieler .........:

Anzahl Tore......:

Weitere Tore (j/):

----------------------------------------

Anzahl der Tore von Spieler 1 : 12

Anzahl der Tore von Spieler 2 : 2

Anzahl der Tore von Spieler 3 : 3

Anzahl der Tore von Spieler 4 : 1

Anzahl der Tore von Spieler 5 : 1

Anzahl der Tore von Spieler 6 : 2

Anzahl der Tore von Spieler 7 : 4

Anzahl der Tore von Spieler 8 : 1

Anzahl der Tore von Spieler 9 : 1

Anzahl der Tore von Spieler 10: 1

Gesamttore....................: 28

Die Aufgabe ist mit Hilfe eines Arrays zu lösen!

Hinweis: Damit die Daten nach dem Ausschalten des Rechners nicht verloren gehen, müßten sie vorher abgespeichert werden. Dies erfolgt an späterer Stelle (Dateiver​arbeitung).

Aufgabe 66:
Vergleichen Sie bitte Aufgaben 63/64 mit der Aufgabe 65. Worin besteht der prinzipielle Unterschied zwischen beiden Aufgaben hinsichtlich der Notwendigkeit, mit einem Array zu arbeiten. 

Tipp: Folgende Aspekte sollten Sie berücksichtigen: „Anzahl der Elemente“, „Reihenfolge der Eingabe“ „Abfragen im Programm (if...)“ und „Wo werden Zwischenergebnisse benö​tigt?“.
Aufgabe 67 (HANDKE, J.):
Würfel-Simulation: Nach der Vorgabe, wie oft gewürfelt werden soll, erzeugt der Computer eine entsprechende Serie von Zufallszahlen und stellt fest, mit welcher relativen Häufigkeit jede einzelne Zahl erscheint.

Hilfe:

{ Zufallszahl zwischen 1 und 6 erzeugen }

ZufallsZahl := random(6)+1; 

Arrays und Type-Vereinbarung:

Type Tabelle_typ = Array[1..4] of real;

var  tab_rech : Tabelle_typ;

Sortierung von Zahlen (Sortieren durch Auswahl, selection sort)

Problemstellung: Es werden Zahlen in beliebiger Reihenfolge eingegeben; die Anzahl ist festgelegt [z. B. 10]. Die Zahlen sollen der Größe nach aufsteigend sortiert und dann ausgegeben werden. Hinweis: Die eingegebenen Zahlen (und die danach sortierten Zahlen) stehen in einem Array.

Lösung: Es wird hier ein einfacher Sortieralgorithmus vorgestellt. Folgende Überlegungen liegen diesem zugrunde:

1. Zunächst muss die kleinste Zahl gesucht und an die erste Stelle des Arrays gebracht  

    werden.

2. Das gleiche wird nun für die zweitkleinste Zahl und dann für die drittkleinste Zahl usw.

    durchgeführt.
Der Überschaubarkeit halber wird hier nur von 4 Zahlen ausgegangen, die nach der Eingabe im Array stehen:


              8             3             5           1 

                    Z[1]                Z[2]                Z[3]             Z[4]


    H
Wenn zunächst die kleinste Zahl an den Anfang des Array geschoben werden soll, müssen alle Zahlen miteinander verglichen werden. Das Schieben bewirkt nun, dass das ursprüngliche Tabellenelement von einer anderen Zahl überschrieben und damit gelöscht wird. Aus diesem Grund existiert noch eine Variable H, eine Hilfsvariable, gleichsam ein Zwischenspeicher. Der Tausch funktioniert nun folgendermaßen:

1. Schritt:

If Z[2] < Z[1] Then Z[2] und Z[1] vertauschen: Z[2]      H,  Z[1]       Z[2],  H       Z[1]


              3             8             5            1 


  3


                    Z[1]                Z[2]                Z[3]              Z[4]


    H
2. Schritt:

If Z[3] < Z[1] Then Z[3] und Z[1] vertauschen: Z[3]      H,  Z[1]       Z[3],  H       Z[1]

hier bleibt alles, wie es war, da die Bedingung nicht zutrifft:

                    Z[1]                Z[2]                Z[3]              Z[4]


    H
3. Schritt:

If Z[4] < Z[1] Then Z[4] und Z[1] vertauschen: Z[4]      H,  Z[1]       Z[4],  H       Z[1]

                    Z[1]                Z[2]                Z[3]              Z[4]


    H
Jetzt steht in der Tat die kleinste Zahl an erster Stelle des Arrays. Das gleiche wird nun für die zweitkleinste Zahl durchgeführt. 
4. Schritt:

If Z[3] < Z[2] Then Z[3] und Z[2] vertauschen: Z[3]      H,  Z[2]       Z[3],  H       Z[2]

                    Z[1]                Z[2]                Z[3]              Z[4]


    H
5. Schritt:

If Z[4] < Z[2] Then Z[4] und Z[2] vertauschen: Z[4]      H,  Z[2]       Z[4],  H       Z[2]

                    Z[1]                Z[2]                Z[3]              Z[4]


    H
Jetzt steht die zweitkleinste Zahl an zweiter Stelle des Arrays. Das gleiche wird nun für die drittkleinste Zahl durchgeführt. 
6. Schritt:

If Z[4] < Z[3] Then Z[4] und Z[3] vertauschen: Z[4]      H,  Z[3]       Z[4],  H       Z[3]

    H
Nun steht die drittkleinste Zahl an der dritten Stelle des Arrays und die höchste Zahl ‘auto​matisch’ an der letzten Stelle: das Array ist aufsteigend sortiert, das Verfahren ist beendet.


selection_sort

Ausgabe der Werte [Z[I]]


Beachten Sie bitte: Hier ist eine innere Schleife (solange bis K > MAX) in eine äußere Schleife (solange bis I > MAX-1) verschachtelt.

Aufgabe 68:
Überprüfen Sie bitte den Algorithmus anhand der Daten auf den vorherigen Seiten in Form eines Schreibtischtests (Schritt 1 bis Schritt 6)! Legen Sie bitte die zusätzlichen Variablen I und K an, MAX ist 4. Codieren Sie dann das Programm!

Aufgaben 69:

Erstellen Sie bitte ein Programm, welches Namen (Datentyp string) sortiert!

Aufgaben 70:

Die Aufgabenstellung von Aufgabe 65 ist dergestalt zu erweitern, dass eine Sortierung ab​steigend nach Tore der einzelnen Spieler vorgenommen wird. Beachten Sie bitte, dass auch die Spielernummer dabei berücksichtigt werden muss!

Aufgabe 71:

Informieren Sie sich hinsichtlich anderer Sortierverfahren (z. B. Bubblesort und Quicksort) und stellen Sie diese Verfahren vor!

Mehrdimensionale Tabellen

Problemstellung:

Es wird eine Statistik über die Filialumsätze (3 Filialen) bezogen auf Artikelgruppen (4 Gruppen) benötigt. Die Datendeklaration, die Eingabeprozedur und eine sehr einfache Ausgabe (ohne Summenbildung) zeigt folgendes (lauffähige) Programmfragment:

program umsatz_art_gr_fil;

Type Tabelle_typ = Array[1..4,1..3] of real;

var i,k  : Integer;

procedure aus(werte_aus : Tabelle_typ);

begin

    {Ausgabe-Fragment, muss noch verbessert werden!)}

    for i:= 1 to 4 do

    begin

         for k:= 1 to 3 do

         begin

            write(werte_aus[i,k]:8:2,'  ');

         end;

         writeln;

    end;

end;

procedure ein;

var wert:Tabelle_typ;

begin

    for i:= 1 to 4 do

    begin

         for k:= 1 to 3 do

         begin

           write('Gruppe ',i:1,' Filiale ',k:1,': ');

           readln(wert[i,k]);

         end;

    end;

aus(wert);

end;

Begin

    writeln('Artikelumsaetze nach Artikelgruppen und Filialen');

    writeln('------------------------------------------------');

    writeln;

    ein;

    readln;

End.

Aufgabe 72:

Analysieren Sie bitte das Programm! Wie wird die 2-dimensionale Tabelle deklariert, wie erfolgt die Ein- und Ausgabe? Beachten Sie bitte auch die Parameterübergabe eines Arrays! Erweitern Sie bitte das Programm so, dass Zeilen- und Spaltensummen ausgegeben werden (was sagen diese Summen aus?). Entwerfen Sie einen geeigneten Bildschirmaufbau für diese Ausgabe!

Aufgabe 73:

Die Aufgabenstellung von Aufgabe 65 ist so zu erweitern, dass eine Auswertung für drei Mannschaften durchgeführt werden kann. Überlegen Sie sich zunächst einen passenden Bildschirmaufbau.

Beispiel einer 3-dimensionalen Tabelle:

Krankenhaus: Zimmerbelegung (nach HERDT/ZIMMERMANN)

Ein Krankenhaus ist in die Stationen A – K aufgeteilt. Jede Station besitzt 3 Stockwerke mit jeweils 30 Einzelzimmern. Das Programm erlaubt die Eingabe jedes neuen Patienten und gibt als Bestätigung die Zimmerbelegung aus.

PROGRAM Zimmerbelegung;

USES    Crt;

VAR     Krankenhaus : Array ['A'..'K', 1..3, 1..30] OF String[20];

        Patient : String[20];

        Station, Weiter : Char;

        StWerk, Zimmer : ShortInt;

PROCEDURE Tabelle_loeschen;

BEGIN

     FOR Station := 'A' TO 'K' DO

         FOR StWerk := 1 TO 3 DO

             FOR Zimmer := 1 TO 30 DO

             Krankenhaus [Station, StWerk, Zimmer] := 'frei';

END;

PROCEDURE Patient_anzeigen;

BEGIN

     GotoXY (5,15); ClrEol;

     Write ('Station ', Station, ' Stockwerk ', StWerk, ' Zimmer ', Zimmer);

     Write (' wird reserviert fuer ', Krankenhaus [Station, StWerk, Zimmer]);

END;

PROCEDURE Tabelle_fuellen;

BEGIN

     REPEAT

           GotoXY (30,7); ClrEol; ReadLn (Station);

           GotoXY (30,8); ClrEol; ReadLn (StWerk);

           GotoXY (30,9); ClrEol; ReadLn (Zimmer);

           GotoXY (30,10); ClrEol; ReadLn (Patient);

IF Krankenhaus [Station, StWerk, Zimmer] = 'frei'

              THEN BEGIN

                        Krankenhaus [Station, StWerk, Zimmer] := Patient;

                        Patient_anzeigen;

                        GotoXY (35,12); ClrEol; ReadLn (Weiter);

                   END

              ELSE BEGIN

                        GotoXY (5,15); ClrEol;

                        Write ('Station ', Station, ' Stockwerk ', StWerk);

                        Write (' Zimmer ', Zimmer, ' BEREITS BELEGT von ');

                        Write (Krankenhaus [Station, StWerk, Zimmer]);

                   END;

     UNTIL Weiter = 'n';

END;

BEGIN

     ClrScr;

     GotoXY (10,5); WriteLn ('----- Krankenhaus verwalten ------');

     GotoXY (10,7); WriteLn ('Station     (A-K):');

     GotoXY (10,8); WriteLn ('StWerk      (1-3):');

     GotoXY (10,9); WriteLn ('Zimmer-Nr. (1-30):');

     GotoXY (10,10); WriteLn ('Patientenname    :');

     GotoXY (10,12); WriteLn ('Weitere Belegung? (j/n)');

     Tabelle_loeschen;

     Tabelle_fuellen;

END.

Aufgabe 74:

Testen Sie bitte das Programm. Beachten Sie die Indizierung: hier bietet Pascal interessante Möglichkeiten. Warum funktioniert das (Tipp: Skalare Datentypen)?

Erweitern Sie das Programm, dass ein Belegungsplan für jede Station ausgegeben werden kann!

3.2 Daten-Verbund (Records)

Bei der Festlegung der Strukturart Array werden Daten vom gleichen Datentyp zu einer größeren Einheit, dem Array (oder der Tabelle) zusammengefasst. Weitaus häufiger liegt aber in der Datenverarbeitung der Fall vor, dass Daten eine Einheit (einen Verbund) bilden, die nicht vom gleichen Datentyp sind. Beispiele solcher Datenstrukturen sind die Informa​tionen, die zur Beschreibung von Kunden oder Artikeln benötigt werden.

Datensatz Kunde:







Kunde



       Kundennummer
Name

Straße
 PLZ     Ort
   Telefon



6 Stellen

25 Stellen
25 Stellen
 5 St.
 30 St.
  14 Stellen

(alle vom Datentyp String, aber unterschiedlicher Länge)

oder:







Artikel



    Artikelnummer
  Bezeichnung
Preis [EK]
Preis [VK]
Mengeneinheit


    6 Stellen, String
  30 Stellen, String
Double
Double
Integer [Code]

Beispiel:

var Artikel_ein: Record

                    art_num   : string[6];

                    art_bez   : string[30];

                    art_pr_ek : real;

                    art_pr_vk : real;

                    art_me    : integer;

                 end;

Der Variablename lautet nun ‘Artikel_ein’; auf die einzelnen Komponenten der Variable (d.h. auf die einzelnen Daten) wird nun mit dem sogenannten Punkt-Operator zugegriffen:

Artikel_ein.art_num

In der Regel ist es aber zweckmäßiger, zunächst den individuellen (benutzerdefinierten) Datentyp zu vereinbaren. Diesem neuen Datentyp wird dann auch ein Name zugeordnet, den die Programmiererin oder der Programmierer selbst wählt.

Beispiel:

Type Artikel_typ = Record

                   art_num: string[6];

                   art_bez: string[20];

                   art_me : integer;

                   art_pr : real;

                 end;

var Artikel_ein, Artikel_aus: Artikel_typ;

Der neu festgelegte Datentyp hat jetzt den Namen 'Artikel_typ' (man hätte ihm selbst​verständlich einen anderen Namen geben könne, z. B. datensatz_artikel). Damit wurde aber noch keine Variable deklariert. Im Programm wird nun eine Variable von diesem speziellen Datentyp deklariert (siehe letzte Zeile des Beispiels, hier werden gleich zwei Variablen dieses neuen Typs vereinbart). 

Das Schreiben des Verbundnamens vor dem Punktoperator bei dem Zugriff auf die Record-Variablen kann mit einem neuen Pascal-Schlüsselwort (with) „abgekürzt“ werden:

procedure rec_aus_fuellen;

begin

  with Artikel_aus do

    begin

       art_num := an;

       art_bez := bez;

       art_me  := me;

       art_pr  := pr;

    end;

end;

Aufgabe 75:

Erstellen Sie bitte ein kleines Testprogramm mit folgender Funktionalität:

1. Die eingegebnenen Kundendaten werden in einem Record gespeichert

2. Diese Daten werden in einen zweiten Record mit gleichem Aufbau übertragen

3. Die übertragenen Kundendaten werden testweise auf dem Bildschirm ausgegeben.

Mit einer einzigen Variablen dieses neuen Typs hat man natürlich zunächst noch nicht viel gewonnen. Records sind der Ausgangspunkt, um viele Daten im Hauptspeicher (z. B. als Array) oder auf Festplatte (z. B. als Datensätze einer Datei) zu speichern. 

Beispiel: Tabelle, deren Elemente aus Records bestehen:

Artikel_tab: array[1..max] of Artikel_typ;

Zugriff:

  Artikel_tab[i].art_num  := num;

  Artikel_tab[i].art_bez  := bez;

  Artikel_tab[i].art_pr   := pr;

Die Strukturart Record kann selbst andere, benutzerdefinierte Strukturen beinhalten. Sind dies Records, wird mit zwei (oder mehreren) Punktoperatoren gearbeitet. (Z.B.: Ein Record-Element stellt ein Datum dar welches als Record aus den Feldern Tag (integer), Monat (Integer) und Jahr (Integer) vereinbart vorliegt.

Aufgabe 76:

Vereinbaren Sie eine Datum-Struktur (TTMMJJJJ) und benutzen Sie diese in einem Record „Zahlungseingang“. Erstellen Sie bitte ein kleines Testprogramm!

Aufgabe 77 (Nach HEIDEMANN):

Aufgabenumfeld: 

Der INFO-Service einer größeren Gebrauchtwagenhalle (max. 200 Stellplätze) soll mittels PCs verbessert werden. 

Details/Anforderungen:

- Telefonische Anfragen nach einem bestimmten Fahrzeugtyp/Aussattung usw. können sofort

   beantwortet werden

- Fahrzeuge können erfasst und wieder gelöscht werden

Vorgaben:

Als Datenstruktur muss ein Array, bestehend aus einzelnen Records (KFZ-Beschreibung) benutzt werden.

Aufgaben:

1. Legen Sie bitte eine geeignete Datenstruktur zur Erfassung der KFZ-Daten fest!

2. Entwickeln Sie bitte eine geeignete Bildschirmmaske!

3. Entwickeln Sie eine modulare Groblösung in Form von Struktogrammen

4. Codieren Sie das Programm in Turbo-Pascal

Hinweis: Hier fehlt noch die Möglichkeit, die eingegebenen Daten abzuspeichern. Dies wird an späterer Stelle realisiert.

Records können über ihren Verbundvariablennamen einander zugewiesen werden. Dies erspart z.B. bei der Codierung eines Sortierprogramms Schreibaufwand, wenn nach Record-Feldern zu sortieren ist:

if Artikel_tab[k].art_bez < Artikel_tab[j].art_bez Then

          begin

            h              := Artikel_tab[k];

            Artikel_tab[k] := Artikel_tab[j];

            Artikel_tab[j] := h;

          end;

Beachte: Artikel_tab und h sind vom gleichen Typ!

Aufgabe 78:

Erweitern Sie bitte das Programm von Aufgabe 77 so, dass nach Datenfeldern, die Sie vorgeben, aufsteigend sortiert werden kann!

Aufgabe 79:

Erstellen Sie bitte ein ähnliches Programm wie das in den Aufgaben 77 und 78 beschriebene zur Verwaltung von Kunden! Erweitern Sie Ihr Programm um die Möglichkeit, eine Kundenliste zu drucken (informieren Sie sich bitte, wie in Pascal gedruckt wird).

Pointer

Werden Daten eines Datenverbunds in Arrays abgespeichert, kann es von Nachteil sein, dass das Array im Vorfeld statisch hinsichtlich seiner Größe definiert werden muss. Es könnte passieren, dass das Array irgendwann einmal zu klein ist, andererseits ist es nicht zweck​mäßig, ein Array stark überzudimensionieren, da dieser Speicherplatz möglicherweise für andere Dinge benötigt wird. Abhilfe schaffen hier Pointer (Zeiger), die eine dynamische „Record-Verwaltung“ erlauben. 

Eine Pointer-Variable zeigt auf eine bestimmte Hauptspeicheradresse, an der ein bestimmter Datensatz zu finden ist. Jedem Datensatz gibt man nun in seine Satzbeschreibung einen Pointer mit, der angibt, an welcher sich der nächste Datensatz befindet (einfache Verkettung), evtl. auch einen Pointer für den vorherigen Datensatz (doppelte Verkettung). Im Vorfeld muss nicht bekannt sein, wie viele Datensätze im Speicher abgelegt werden sollen; solange noch Speicherplatz vorhanden ist, können neue Datensätze eingefügt werden. Die Hauptarbeit bei der Pointerverwaltung übernimmt das System, von der Programmierung her muss nur dafür gesorgt werden, dass die Verkettung tatsächlich stimmt. Dass Pointervariablen tatsächliche Hauptspeicheradressen enthalten ist nicht direkt sichtbar und die phyischen (echten) Speicheradressen müssen den ProgrammiererInnen nicht bekannt sein.

Beispiel (nach HANDKE):



type Zeiger = ^DatenTyp;


     DatenTyp = record

                VorName,

                NachName: string [30];

                Vorgaenger, Nachfolger: Zeiger;

                end;


var  Erster_Datensatz, Laufender_Datensatz,

     Eben_Eingegebener_Datensatz: Zeiger;


Das Arbeiten mit Pointer

Neuer Speichrplatz für einen neuen Daten-Record während des Programmlaufs schafft die Anweisung NEW (<Zeiger_var>). <Zeiger_var> erhält dann als „Wert“ dann auch eine Speicheradresse. Mit DISPOSE (<Zeiger_var> wird der Speicherbereich wieder freigegeben. Einer Zeiger-Variablen kann kein direkter Wert (Speicheradresse) zugewiesen werden (dies soll ja das System verwalten); es ist aber (natürlich) möglich, einer Zeigervariablen den Inhalt (=Speicheradresse) einer anderen Zeigervariablen zuzuweisen: 

Eben_Eingegebener_Datensatz := Laufender_Datensatz;

Soll ein Zeiger nirgendwohin zeigen, erfolgt dies mit NIL:

Laufender_Datensatz^.Nachfolger := nil;

Die Einzelkomponenten von Records, auf den Zeiger zeigen, werden einander folgendermaßen zugeordnet (Beispiel):

Laufender_D^.NachName := Eben_Eingegebener_D^.NachName;

(Hinweis: Die With-Anweisung funktioniert auch hier!)

Ganze Records einander zuordnen:

Erster^ :=  Laufender^;

Aufgabe 80:

Analysieren Sie bitte das Programm auf der nächsten Seite (von HANDKE, J.). Stellen Sie bitte die Datenstruktur im Hauptspeicher für vier Namenseingaben in folgender Form (hier unvollständig) grafisch dar: 






Zeigt auf einen neuen Pointer

Aufgabe 81:

Informieren Sie sich über die Möglichkeiten der Verarbeitung dynamischer Strukturen (Internet/Literaturliste dieser Unterlagen/sonst. Literatur) und stellen Sie bitte wichtige Möglichkeiten grafisch dar (evtl. Präsentation).

program NamensListe;

                    { Aufbau und Ausdruck einer dynamischen Klassenliste }

uses crt;

type Zeiger = ^DatenTyp;

     DatenTyp = record

                VorName,

                NachName: string [30];

                Vorgaenger,

                Nachfolger: Zeiger;

                end;

var  Erster_Datensatz, Laufender_Datensatz, Eben_Eingegebener_Datensatz: Zeiger;

     i: integer;

begin

   ClrScr;

   new ( Erster_Datensatz );                     { Eingabe des ersten Datensatzes }

   Erster_Datensatz^.Vorgaenger := nil;

   write ('   Eingabe des Nachnamens ( Abbruch mit "-" ) :     ');

   readln ( Erster_Datensatz^.NachName );

   write ('   Eingabe des Vornamens :                          ');

   readln ( Erster_Datensatz^.VorName );

   Eben_Eingegebener_Datensatz := Erster_Datensatz;

   repeat

         for i:=1 to 3 do writeln;

         new ( Laufender_Datensatz );            { Eingabe des n„chsten Datensatzes }

         Laufender_Datensatz^.Vorgaenger := Eben_Eingegebener_Datensatz;

         Eben_Eingegebener_Datensatz^.Nachfolger := Laufender_Datensatz;

         write ('   Eingabe des Nachnamens ( Abbruch mit "-" ) :     ');

         readln ( Laufender_Datensatz^.NachName );

         write ('   Eingabe des Vornamens :                          ');

         readln ( Laufender_Datensatz^.VorName );

         Eben_Eingegebener_Datensatz := Laufender_Datensatz;

   until Laufender_Datensatz^.NachName = '-';

   Laufender_Datensatz^.Nachfolger := nil;

   ClrScr;                                       { Ausgabe aller Datens„tze }

   Laufender_Datensatz := Erster_Datensatz;

   while Laufender_Datensatz^.Nachfolger <> nil do

         begin

         write ( Laufender_Datensatz^.NachName:30, Laufender_Datensatz^.VorName:30 );

         writeln;

         Laufender_Datensatz := Laufender_Datensatz^.Nachfolger;

         end;

end.

3.3 Set

Die Datenstruktur Set (Menge) ist eine Sammlung von Konstanten des gleichen Datentyps. Sie lassen sich geschickt für Bedingungsprüfungen einsetzen wobei geprüft wird, ob ein be​stimmter Wert in der Menge enthalten ist.

Programmbeispiel:

program datentypen_2;

Grundmenge festlegen: }

Type Tage    = Set of 0..100;

Type zeichen = Set Of 'a'..'f';

{ Teilmenge festlegen:  }

const ta : Tage     = [1..31];

      ze : zeichen = ['a', 'b'];

var   t  : shortint;

      z  : char;

      ein: string;

begin

      t := 20;

{ Mengenzugehörigkeit prüfen: }

      if (t in ta) then writeln('Tag o.k.!');

      t := 123;

      if (t in ta) then writeln('Tag o.k.!')

      else writeln ('Jetzt aber nicht o.k.!');

      z := 'b';

      if (z in ze) then writeln('Buchstabe falsch!');

read(ein);

end.

Aufgabe 82:

Erklären Sie die Syntax und die Funktionsweise dieses Programms bezüglich der Datenstruktur Set!

3.4 Dateien (Files)

Pascal kennt verschiedene Dateitypen, von denen hier zweit Typen vorgestellt werden.

Text-File

Bei diesen Dateien dürfen die einzelnen Datensätze unterschiedlich lang sein. Am Ende eines Datensatzes wird eine End Of Line – Markierung (ASCII 13 und 10) gesetzt, End of File wird mit ASCII 26 markiert.

Beispiel einer Textdatei (nach KASSERA, W. und V.):

Program Text_File_Test;

VAR  T_File   :Text;


     i        :integer;

     ein      :string[60];

begin

   assign(T_File, 'test-d.txt');


   Rewrite(T_File);

   For i := 1 to 10 Do

      writeln(T_File,i:2,'. Testzeile');

   close(T_File);

   writeln('Datei geschrieben!');

   writeln;

   writeln('Datei wird jetzt gelesen!');

   assign(T_File, 'test-d.txt');

   Reset(T_File);


   While not EOF (T_File) do

      begin

        readln(T_File,ein);

        writeln(ein);

      end;

   close(T_File);

   writeln('Lesen beendet!');

   readln(i);

end.

Folgendes steht in der Datei:

 1. Testzeile

 2. Testzeile

 3. Testzeile

 4. Testzeile

 5. Testzeile

 6. Testzeile

 7. Testzeile

 8. Testzeile

 9. Testzeile

10. Testzeile

Beachte: Die Zeile 

writeln(T_File,i:2,'. Testzeile');

bewirkt, dass die Integerzahl nicht im internen Format, sondern als lesbares Zeichen in der Datei steht (das ist nicht immer so, s. später!).

Aufgabe 83:

Informieren Sie sich über weiter Möglichkeiten/Anweisungen hinsichtlich der Textdai-Verarbeitung!

Aufgabe 84:

Erstellen Sie bitte ein Programm, welches Pascal-Quellprogramme mit Zeilennummern ausdruckt!

Files mit Datensätzen gleicher Länge

Files dieser Art können aufgrund ihrer konstanten Satzlänge im Direktzugriff verarbeitet werden. Das bedeutet, dass im Gegensatz zu den Textdateien, die nur sequentiell (von Anfang bis Ende) verarbeitet werden, einzelnen Datensätze direkt mitten aus der Datei gelesen werden können, ohne dass die vorherigen Sätze gelesen wurden: Das System kann sich (mit Hilfe der SEEK-Anweisung) die Satzposition „relativ“ von Dateianfang aus berechnen (daher auch der Name: relative Dateien). Deklariert wird eine solche Datei mit dem Zusatz FILE OF.... Steht hinter OF ein Verbundvariabletyp, ist die konstante Datensatzlänge ent​sprechend der Gesamtlänge des Records festgelegt:

VAR  R_File   : File of Artikel_typ;

Aufgabe 85:

Informieren Sie sich über die Möglichkeiten/Anweisungen hinsichtlich der „File-Of“-Dateiverarbeitung!

Beispielprogramm:

Program Persdruck;

 Const dateiname       = 'c:\PDAT';

 Type  drucktyp_datum  = string[8];

       zeichen6        = string[6];

       satz            = record nummer   : string[6];

                                name     : string[20];

                                eintritt : string[6];

                                (*sonstige Daten*)

                         end;

 Var   datei                  : file of satz;

       person                 : satz;

       i                      : integer;

       druckjahr              : drucktyp_datum;

       a, b, c                : integer;

function druckaufbereitung_datum (conv: zeichen6): drucktyp_datum;

 var    za : drucktyp_datum;

     k, i  : integer;

 begin

    za := '        ';

    za[8] := conv[6];

    za[7] := conv[5];

    za[6] := '.';

    za[5] := conv[4];

    za[4] := conv[3];

    za[3] := '.';

    za[2] := conv[2];

    za[1] := conv[1];

    if za[1] = ' ' then za[1] := '0';

    druckaufbereitung_datum := za

 end;

 procedure kopf;

 begin

  writeln ; writeln ;

  writeln ('PERSONALLISTE');

  writeln ;

  writeln ('NAME,VORNAME           PERS.-   EINTR.-');

  writeln ('                       NUMMER   DATUM  ');

  writeln ;

  end;

 procedure druck;

  begin

    with person do

    begin

      druckjahr := druckaufbereitung_datum (eintritt);

      name := name + '                          ';

      writeln (name:20 ,'   ',nummer,'   ', druckjahr);

    end;

  end;

 procedure einlesen;

 begin

  while not EOF (datei) do

    begin

    read (datei, person);

    with person do

      begin

       druckjahr := druckaufbereitung_datum (eintritt);

      end;

      druck;

     end;

 end;

(*=========================================================*)

 begin

   assign(datei, dateiname);

   reset(datei);

   kopf;

   einlesen;

   close (datei);

   readln(i);

 end.

Aufgabe 86:

Analysieren Sie diese Programm! Testen Sie es. Zuvor müssen Sie allerdings ein Programm zur Abspeicherung von Testdaten auf der Festplatte erstellen.

Aufgabe 87:

Erweitern Sie das Programm von Aufgabe 78 dergestalt, dass ein Abspeichern der Daten auf Festplatte und ein Wiedereinlesen der Daten möglicht ist!

Aufgabe 88 Werkzeugvermietung (nach AKA, Prüfung DV-Kaufleute, Codierung)

Aufgabenumfeld:

Ein großer überregionaler Heimwerkermarkt erstellt Rechnungen für vermietete Werkzeuge. Der Mietpreis ist abhängig von der Art des Werkzeuges, dem Fabrikat und der Mietdauer. Folgende fünf Werkzeugarten werden unterschieden:

Werkzeugart 1:
Bohrhammer/-maschine

Werkzeugart 2:
Gartenabfallschredder

Werkzeugart 3:
Wasserpumpe

Werkzeugart 4:
Betonmischer

Werkzeugart 5:
Motorsägen

Innerhalb jeder Werkzeugart gibt es drei verschiedene Fabrikate (F1, F2, F3). 

Dateien:

Die erforderlichen Daten stehen in zwei Dateien; alle Daten (auch die nummerischen Daten) liegen in einem alphanumerischen Format (string) vor. In der Datei 1 stehen die Werkzeugbezeichnung, das Fabrikat, die (bisher aufgelaufene) Summe der Leihtage und der Preis pro Leihtag. Diese Datei ist nach Programmstart  in eine Tabelle mit folgendem Aufbau in den Arbeitsspeicher zu laden (Preis je Leihtag: Komma ist nicht gespeichert, die letzten beiden Ziffern sind die Nachkommastellen):

Werkzeugart 1 
Werkzeugart 2 
......



Werkzeugart 5

F1
F2
F3


Werkzeugbez.     Fabrikat


Summe Leihtage

Preis pro Leihtag

  1-5

       6-28


29-32



33-37

Die vermieteten Werkzeuge stehen in einer Datei (Datei 2) mit folgendem Satzaufbau:

Satzposition

Datenfeld



Var.-Name

 1


Werkzeugart

 2-3


Fabrikat

 4-23


Mietername

24-29


Datum der Ausleihe (TTMMJJ)

30-84


Adressangaben

85-87


Anzahl Leihtage

Diese Datei wird sequentiell gelesen. Jeder Datensatz repräsentiert ein Werkzeug, welches 

von einem Mieter ausgeliehen wurde. Der gleiche Mieter kann mit dem gleichen Werkzeug oder einem anderen Werkzeug natürlich mehrmals in der Datei stehen; selbstverständlich kann das gleiche Werkzeug auch von verschiedenen Mietern ausgeliehen werden. 

Ausgabe:

Je Mieter und Werkzeugart ist eine Rechnung mit folgendem Daten auszugeben:

Kopfdaten der Rechnung

Name des Mieters: XXXXXXXXXXXXXXXXXXXX

Ausgeliehen am
: XX.XX.XX

Werkzeugart

: XXXXXXXXXXXXXXX

Fabrikat


: XXXXXXXXXXXXX

Anzahl Leihtage
: XXX

Preis pro Leihtag
: XXX

Rechnungsbetrag
: XXXXX,XX

Lösungsschritte:

1. Entwerfen Sie bitte ein geeignetes Rechnungsformular!

2. Definieren Sie die erforderlichen Datensätze und die Tabelle zur Aufnahme der Werk​zeugangaben!

3. Erstellen Sie ein Struktogramm aus dem hervorgeht, wie die Daten aus der Datei 1 in den Hauptspeicher gelesen werden (Tabellenindizes beachten und angeben)!

4. Erstellen Sie bitte ein Struktogramm für das sequentielle Lesen der Datei 2 und die Ausgabe der jeweilige Abrechnung!

5. Erstellen Sie die Struktogramme zu den noch fehlenden Programmteilen!

6. Codieren Sie das komplette Programm! Berücksichtigen Sie dabei, dass das Feld „Summe Leihtage“ der Datei 1 zu aktualisieren ist (Tabelle zurückspeichern)

Hinweise: 

Pro ausgeliehenem Werkzeug (Datei 2) wird eine Abrechnung ausgegeben.

Folgende Themenbereiche der Datenverarbeitung werden durch diese Aufgabe mitan​gesprochen:

· Datenflussplan

· Top-Down-Vorgehensweise, modulare Programmierung

· Hierarchisch strukturiertes Programm, Programmorganisationsplan

Machen Sie sich bitte klar, dass dieses Programm nur getestet werden kann, wenn die entsprechenden Eingabedateien vorliegen. Entweder erstellen Sie zu diesem Zweck noch zwei Erfassungsprogramme oder Sie geben die Daten mit Hilfe eines Editors ein (Satzlängen bzw. Satzfelder beachten!).

Aufgabe 89:

Erweitern Sie das Programm von Aufgabe 88 dergestalt, dass Rechnungen für einzelne, vorher festgelegte Mieter ausgegeben werden!

Direktzugriff mit SEEK, Beispiel:

SEEK(X_FILE, i);

X_FILE: Datei-Variable, i: Typ integer, Satznummer, Achtung die Positionszählung beginnt ab 0!

Aufgabe 90:

Erstellen Sie bitte ein Programm zur Artikelverwaltung (mit Dateiabspeicherung). Legen Sie bitte eine Artikelnummer fest, aus der sich durch einfache Berechnung eine Satznummer berechnen lässt. Ermöglichen Sie bitte einen Direktzugriff auf die einzelnen Artikel​datensätze, wenn die Artikelnummer (als Schlüssel) eingegeben wurde. Überlegen Sie, wie sich das Löschen von Sätzen realisieren lässt. Welche weiteren sinnvollen Funktionen sollte das Programm noch haben. Erstellen Sie diese!

Programmlogik für die satzweise Verarbeitung von Dateien: Normierte Programmierung (Gruppenwechsel)

In einem Betrieb soll aus den Abrechnungsdaten der Rechnungsschreibung ein Verkaufs-bericht erstellt werden. In dem Bericht sollen die Umsätze je Artikelnummer mit Zwischen-summen je Kundennummer und der Gesamtumsatz der Firma ausgewiesen werden:






Beispiel: Umsatzliste (aus einem „alten“ COBOL-Lehrbuch (SCHWANKE, COBOLL-Programmierung M&T-Verlag)

Heute werden derartige Probleme i. d. R. mit den entsprechenden Select-Anweisungen in SQL gelöst, lange Zeit aber musste die Lösung „von Hand“ programmiert werden, wenn kein entsprechender Programmgenerator zur Verfügung stand. 

Die Satzbeschreibung der Datei entspricht den Spalten der Liste bis einschließlich RECH-NR; sortiert ist die Datei aufsteigend nach Artikelnummer.

Aufgabe  Entwicklung der entsprechende Programmlogik nach folgendem Schema:








Lösung: Umsatzliste

Gruppierwörter:


GW-ALT: 


GW-NEU:

Beispiel: 1.
GW-NEU-KU-NUM: Die Kundennummer des gerade eingelesenen Satzes

Beispiel:2.
GW-ALT-ART-NUM: Die Artikelnummer des gerade verarbeiteten Satzes

Hinweis: Solange GW-NEU = GW-ALT, hat kein Gruppenwechsel stattgefunden. Das bedeutet, dass zum gleichen Artikel in der Datei ein weiterer Datensatz mit der gleichen Kundennummer gefunden wurde.

Logik (orientiert sich an DIN 66220 und 66260, diese Vorschriften stellen die Logik allerdings als PAs dar und sind etwas anders strukturiert)

Hinweis: Selbstverständlich sind auch andere Lösungen möglich, diese DIN-Vorgaben hatten (u. a.) den Sinn, ein Lösungsschema zu standardisieren, damit komplexere Gruppenwechsel-Programmen (z. B. 5 Eingabedateien, 4-stufiger Wechsel) sicher bzw. fehlerfrei gelöst werden können und eine Programmwartung nicht nur vom Ersteller des Programms sondern von je​dem, der die DIN kennt, in einem zeitlich vertretbaren Rahmen durchgeführt werden kann.

Hauptprogramm: nächste Seite

UP-KOPF






UMSATZLISTE-DRUCKEN

Vorlauf: 

GW-ALT = LOW VALUE, Gesamtumsatz = 0, SZ = 1, ZZ = 7

Datei öffnen, Drucker öffnen

Ersten Datensatz lesen, GW-NEU aufbauen






     EOF?


ja







nein


GW-NEU = HIGH VALUE


./.



Solange GW-NEU <> HIGH VALUE



Vorlauf Artikelverarbeitung: Artikelumsatz = 0



GW-NEU-ART-NUM

Druckbereich der Pos.-Zeile

                        GW-ALT-ART-NUM = GW-NEU-ART-NUM



Solange GW-NEU-ART-NUM = GW-ALT-ART-NUM




Vorlauf Kundenverarbeitung: Kundenumsatz = 0




GW-ALT-KU-NUM = GW-NEU-KU-NUM




Solange GW-NEU = GW-ALT




Einzelverarbeitung:
Kundenumsatz addieren





Positionszeile druckaufbereiten








ZZ > 60






ja



nein





  UP-KOPF



./.






Positionszeile drucken, ZZ = ZZ + 1





ART-NUM im Druckbereich löschen





Nächsten Satz lesen, GW-NEU aufbauen








EOF






ja


         

nein





GW-NEU = HIGH VALUE

./.



Nachlauf KU:
Kundenumsatzzeile druckaufbereiten und drucken





Artikelumsatz aufaddieren


Nachlauf ART: Artikelsummenzeile druckaufbereiten und drucken


 

 Gesamtumsatz aufaddieren

Ende-Verarbeitung: 
Gesamtumsatz druckaufbereiten und drucken




Datei und Drucker schließen



Hinweis: In den DIN-Vorschriften sind die im Struktogramm kursiv gedruckten Verarbei​tungsschritte als Unterprogramme realisiert (Vorlauf, Gruppenanfangsverarbeitungen, Nachlauf, Einzelverarbeitung). Dadurch wird bei komplexeren Aufgaben das gesamte Programm vom Aufbau her sehr klar strukturiert.

Aufgabe 91:

Codieren Sie bitte diese GW-Ablauflogik in PASCAL!

Aufgabe 92:

Für Spezialisten und Enthusiasten:

Schauen Sie sich die Beispielaufgaben von DIN 66220 und 66260 an. Hier sind neben der sequentiellen Zugriffsweise auch Direktzugriffsdateien zu verarbeiten. Überführen Sie die PAs in Struktogramme, legen Sie geeignete Testdateien fest und führen Sie Schreibtischtests durch. Codieren Sie Ihre Lösung und testen Sie das Programm! (Achtung: Diese Aufgabe ist zeitlich kaum an einem Tag zu bewältigen, wenn man das Problem zum ersten Mal angeht. Vielleicht können Sie sie in Gruppenarbeit mit KollegInnen projektorientiert lösen). Falls Sie Erfahrungen mit SQL haben: Wie sieht dort die entsprechende Lösung aus?

Ende des PASCAL-Kurses!

ART-NUM   KU-NUM





Vorlauf


Solange bisEOF


Vorlauf Haupt-Gruppe 


Solange bis neuer Artikel


Vorlauf Unter-Gruppe 


Solange bis neuer Kunde


usw.





�





Dies ist erforderlich, weil das Datum  ohne Punkte in der Datei steht!





Lese-Schleife bis EOF





Datei vorbereiten, dass vom 1. Datensatz aus gelesen werden kann





Datei schließen





Schreiben von Daten auf die Festplatte (unter Angabe des Datei�namens)





Mit Rewrite wird die neue Datei auf der Fest�platte angelegt, eine evtl. existierende Datei mit gleichem Namen wird überschrieben (dies lässt sich auch vermeiden).





Zuweisung eines Datei�namens (auf der Festplatte) an die Dateivariable





Festlegen des Dateityps „Text“ mit einer Dateivariablen





Peter Meier	  PTR-Vor.	





Mit Hilfe dieser Zeigervariablen werden im Programm dann die Zuordnungen zum Aufbau der Struktur vorgenommen.





Jeder Datensatz enthält die Speicheradresse seines Vorgängers und seines Nachfolgers.





Eine Zeiger-Variable ist untrennbar mit dem ver�bunden, worauf sie zeigt: Nur durch den Daten�typ-Bezug kann dass System berechnen, auf welche Adresse der Nachfolger zeigen soll, da diese von der Record-Länge abhängig ist.





Zeigersymbol





ART-NUM   KU-NUM





Kopfzeilen druckaufbereiten (Datum, Seitenzähler berücksichtigen)


Blattvorschub


Kopf drucken


SZ = SZ + 1


ZZ = 7




















UP-KOPF








SEITE  

